

A GUIDE TO PROGRAMMING IN LEVEL II BASIC



BRUCE PRESLEY

ANSWER KEY



DISTRIBUTED BY



VAN NOSTRAND REINHOLD COMPANY

NEW YORK

CINCINNATI

TORONTO

LONDON

MELBOURNE

CHAPTER 1

2. 10 PRINT 8*(19-4)
 20 PRINT 1*2+2*3+3*4+4*5
 30 PRINT 1+ 1/10 + 1/100
 40 PRINT 10-20+30-40+50-60
4. 28.6667
 25
 48
 44.3333
 Out of DATA in 10
6. S
 I
 X!
 6
 6
 Out of DATA in 20
8. 10 INPUT "POUNDS, INCHES";P,N
 20 PRINT P/N;" POUNDS PER INCH."
10. >RUN
 TOTAL PRICE = \$0 .89
12. >PRINT .05
 .05

 >PRINT .005
 5E-03

 >PRINT 123456789
 123456789

 >PRINT 1234567890
 1234567890

 >PRINT -.065
 -.065

 >PRINT 4.62E5
 462000

 >PRINT .006E-3
 6.6E-05

 >PRINT -4.900
 -4 900

```

14.  10 PRINT 3+5+7

16.  10 READ A,B
     20 PRINT (7*A+10*B)/(2*A*B)
     30 PRINT ((1/2)*A)/(A-B)
     40 DATA 10,7

18.  10 INPUT "HOW MANY PIECES DID YOU EAT";P
     20 C = 375*P
     30 M = C/100
     40 PRINT "YOU MUST RUN";M;"MILES."

20.  10 INPUT "VICTIMS NAME";V$
     20 INPUT "INSULTING WORD";W$
     30 PRINT V$;" IS A ";W$;"-HEAD"
     40 PRINT "      OR"
     50 PRINT V$;" IS A ";W$;"-BREATH"
     60 GOTO 10

22.  10 READ R
     20 PRINT:PRINT
     30 A=3.141592654*R*R
     40 PRINT "AREA OF CIRCLE =";A
     50 GOTO 10
     60 DATA 5,3,8

24.  10 C = 3E10
     20 M = 2E11/(C*C)
     30 N=M/(.1)
     40 PRINT N;"HAIRS ARE REQUIRED."

26.  10 READ A, B, C
     20 PRINT "SEMIPERIMETER OF FIRST TRIANGLE ";(A+B+C)/2;"CM."
     30 READ A, B, C
     40 PRINT "SEMIPERIMETER OF SECOND TRIANGLE ";(A+B+C)/2;"FT."
     50 DATA 13, 8, 11, 21, 16, 12

28.  10 INPUT "SALE PRICE";P
     20 INPUT "DISCOUNT RATE";D
     30 D = D/100
     40 PRINT "THE ORIGINAL PRICE WAS $";P/(1-D)
     50 GOTO 10

```

30. 10 INPUT "ITEMS NAME";N\$
 20 INPUT "WHAT IS ITS PRICE";P
 30 T=.07*P
 40 PRINT N\$;" HAS A TAX OF \$";T;"AND COSTS \$";P+T
 50 GOTO 10
32. 10 D = 2.6
 20 A1 = D*D
 30 A2 = 3.141592654*(D/2)*(D/2)
 40 A3 = A1-A2
 50 PRINT "AREA LEFT =";A3;"SQUARE CENTIMETERS."
34. 10 INPUT "HOW MANY NUMBERS";N
 20 INPUT "WHAT ARE THE NUMBERS";A,B,C,D,E
 30 V = (A+B+C+D+E)/N
 40 PRINT "AVERAGE IS";V
 50 GOTO 10

Note: If less than five numbers are in the set, 0's are used to
 fill in the set

CHAPTER 2

2.

```
10 INPUT "3 NAMES";A$,B$,C$
20 IF A$ > B$ AND A$ > C$ THEN PRINT A$ : GOTO 10
30 IF B$ > C$
    THEN PRINT B$
    ELSE PRINT C$
40 GOTO 10
50 GOTO 10
```
4.

```
10 INPUT X$
20 IF X$ = "BIGWOW"
    THEN PRINT "!!!!!!"
    ELSE PRINT "??????"
30 GOTO 10
```
6.

```
10 INPUT X
20 IF X < -24 OR X > 17 THEN PRINT "NOT BETWEEN"
30 GOTO 10
```
8.

```
10 INPUT X
20 IF X > 25 AND X < 75
    THEN PRINT "IN THE INTERVAL"
    ELSE PRINT "NOT IN THE INTERVAL"
30 GOTO 10
```
10.

```
10 FOR I = 1 TO 3
20     PRINT "*****"
30 NEXT I
```
12.

```
10 FOR I = 4 TO 84 STEP 10
20     PRINT I
30 NEXT I
```
14.

```
>PRINT 143*74
    10582

>PRINT (53+72+81+76)/4
    70.5

>X=4:Y=5:PRINT"X[Y=";X[Y,"Y[X=";Y[X
X[Y= 1024      Y[X= 625
```
16.

```
10 PRINT "*****"
20 FOR I = 1 TO 6
30     PRINT " ***"
40 NEXT I
50 PRINT "*****"
```

18. Line: <10><20><50><80><10><20><30><60><80>
<10><20><30><40><70><80> End
20. 10 INPUT "HOW OLD ARE YOU ";A
20 IF A >= 16
 THEN PRINT "YOU ARE OLD ENOUGH TO DRIVE A CAR "
 ELSE PRINT "YOU MUST WAIT";16-A;"YEARS TO DRIVE"
22. 10 FOR I = 1 TO 10
20 READ N\$,A: REM GET NAME , CONTRIBUTION
30 PRINT : PRINT "DEAR ";N\$;" , "
40 PRINT " THANK YOU FOR YOUR GENEROUS CONTRIBUTION"
50 PRINT "OF";A;"DOLLARS TO MY ELECTION CAMPAIGN. MAYBE"
60 PRINT "NEXT YEAR WE WILL HAVE BETTER LUCK!"
70 PRINT TAB(30);"SINCERELY,"
80 PRINT : PRINT TAB(30);"SMILEY R. POLITICO"
90 PRINT
100 NEXT I
110 DATA RICH BRYBURY,25000,SWINDLING SCHWABB,1000
120 DATA SEMORE FUNDS,500,KILLER KIDDER,50
130 DATA LARS CADET,150,DR. CRABAPPLE,675,EVE,5000
140 DATA GARY BONEHEAD,15,SIR ROBIN,768,ADAM,2800
24. 10 FOR I = 1 TO 10
20 RESTORE
30 INPUT "WHAT MODEL DID YOU BUY";M
40 READ D : REM GET A BAD MODEL NUMBER
50 IF D = -1 THEN PRINT "YOUR CAR IS FINE!":GOTO 80
60 IF D = M THEN PRINT "YOU GOT A LEMON!":GOTO 80
70 GOTO 40 : REM GO CHECK NEXT MODEL NUMBER
80 NEXT I
90 DATA 102,780,119,220,189,195,-1 : REM -1 SHOWS END OF DATA
26. 10 FOR X = 5 TO 25 STEP 2 : REM STEPPED 2 FOR ODDS ONLY
20 IF X/3 > 500 AND X/2 + 3*X + 2 < 700 THEN PRINT X
30 NEXT X

```

28.  1 REM    C1 = PRICE PER SHARE WHEN PURCHASED
      2 REM    C2 = PRICE PER SHARE WHEN SOLD
      3 REM    T1 = TOTAL VALUE OF STOCK PURCHASED
      4 REM    T2 = TOTAL VALUE OF STOCK SOLD
      5 REM    P = PROFIT
      6 REM    N = NUMBER OF SHARES OF STOCK
      7 REM
      8 REM
     10 READ A$,N,C1,C2
     20 T1=N*C1
     30 T2=N*C2
     40 P=T2-T1
     50 PRINT "FOR ";A$;
     60 IF SGN(P)=1
           THEN PRINT "THE PROFIT WAS $";P
           ELSE PRINT "THE LOSS WAS $";-P
     70 GOTO 10
     80 END
     90 REM    DATA SECTION
    100      DATA "C.T.",200,85.58,70.82
    110      DATA "A.A.S.",400,35.60,47.32

30.  10 INPUT "THE AMOUNT OF THE LOAN";L
     20 INPUT "LENGTH OF THE LOAN IN YEARS";Y
     30 I = .02 : REM INTEREST RATE PER MONTH
     40 M = Y * 12
     50 P = (I*L) / ((1-(1+I)(-M)))
     60 PRINT USING "THE MONTHLY PAYMENT IS #.##";P
     70 PRINT USING "TOTAL AMOUNT PAID WILL BE ###.##";P*M

```


CHAPTER 3

2. 10 FOR I = 1 TO 3
 20 S = S + RND(0)
 30 NEXT I
 40 PRINT S

4. 10 N = RND(8) - 4
 20 INPUT "GUESS A NUMBER FROM -3 TO 4";G
 30 IF N = G
 THEN PRINT "CORRECT"
 ELSE PRINT "INCORRECT IT WAS ";N
 40 GOTO 10

6. 10 INPUT N
 20 FOR I = 1 TO N
 30 S = S + RND(0)
 40 NEXT I
 50 PRINT "SUM IS ";S
 60 PRINT "N/2 IS ";N/2
 70 GOTO 10

8. 10 FOR I = 1 TO 4
 20 PRINT "1234567890";
 30 NEXT I
 40 PRINT:PRINT
 50 PRINT "*";TAB(8);"9";TAB(14);"?";TAB(20);"-";
 60 PRINT TAB(28);"!";TAB(32);")";TAB(39);"*"

10. 10 CLS
 20 REM DRAW BOX
 30 FOR X = 42 TO 84
 40 SET(X,17) : SET(X,33)
 50 NEXT X
 60 FOR Y = 17 TO 33
 70 SET(42,Y) : SET(84,Y)
 80 NEXT Y
 90 REM
 100 REM PRINT MESSAGE
 110 PRINT@ 470,"TODAY ONLY!";
 120 PRINT@ 538,"UNCLE BILL'S";
 130 PRINT@ 602,"WHAMBURGERS";
 140 PRINT@ 670,"ONLY \$0.59";
 150 FOR I = 1 TO 100 : NEXT I : REM DELAY
 160 REM
 200 REM ERASE MESSAGE (MAKE APPEAR TO FLASH)
 210 PRINT@ 470," ";
 220 PRINT@ 538," ";
 230 PRINT@ 602," ";
 240 PRINT@ 670," ";
 250 FOR I = 1 TO 100 : NEXT I : REM DELAY LOOP
 260 GOTO 100

```

12.  10 FOR I = 1 TO 10
      20   PRINT RND(18) + 7;
      30 NEXT I : PRINT

```

14A. >RUN

```

          6 * 3 =? 9
YOU ARE WRONG. TRY AGAIN.
          6 * 3 =? 18
CORRECT

```

```

14B. 10 CLS : PRINT@384,
      20 A = RND(10)
      30 B = RND(10)
      40 PRINT TAB(12);A;"*";B;"=";
      50 INPUT C
      60 IF C = A * B THEN 90
      65 W = W + 1
      67 IF W = 3 THEN PRINT "LEARN THE MULTIPLICATION TABLE":END
      70 PRINT "YOU ARE WRONG. TRY AGAIN."
      85 GOTO 40
      90 PRINT TAB(12);"CORRECT"

```

```

14C. 5  FOR I = 1 TO 5
      10   CLS: PRINT@384,
      15   W = 0 : REM RESET TO GIVE THREE CHANCES FOR EACH QUESTION
      20   A = RND(10)
      30   B = RND(10)
      40   PRINT TAB(12);A;"*";B;"=";
      50   INPUT C
      60   IF C = A * B THEN 90
      65   W = W + 1
      67   IF W = 3 THEN PRINT "LEARN THE MULTIPLICATION TABLE":END
      70   PRINT "YOU ARE WRONG. TRY AGAIN."
      75   F = 1 : REM 1 WILL INDICATE AN INCORRECT ANSWER WAS GIVEN
      80   GOTO 40
      90   PRINT TAB(12);"CORRECT"
      92   R = R + 1
      95   FOR J = 1 TO 300 : NEXT J : REM TIME DELAY
      100  NEXT I
      110 IF F <> 1 THEN PRINT "NICE GOING, HOTSHOT"

```

```

16.  10 INPUT "INTEGER > 1 PLEASE";N
      20 FOR X = 2 TO N-1
      30   IF N / X = INT(N / X) THEN 60
      40 NEXT X
      50 PRINT "THAT INTEGER IS PRIME." : END
      60 PRINT "THAT INTEGER IS NOT PRIME."

```

```

18. 10 INPUT "WOULD YOU LIKE TO GO FIRST ";A$
    20 IF A$ = "NO" THEN 70
    30 INPUT "INPUT A NUMBER FROM 3 - 12";N
    40 IF N < 3 OR N > 12 THEN PRINT "NOT IN RANGE OF 3-12":GOTO 30
    50 S = S + N : PRINT "SUM IS NOW ";S
    60 IF S > 100 THEN PRINT "YOU WIN!!" : END
    70 C = RND(10) + 2
    80 PRINT "MY PICK IS ";C
    90 S = S + C : PRINT "THE SUM IS NOW ";S
    100 IF S > 100 THEN PRINT "I WIN!!!" : END
    110 GOTO 30

20. 10 INPUT "A POSITIVE INTEGER, PLEASE";N
    20 IF N <> INT(N) THEN PRINT "YOUR NUMBER WAS NOT AN INTEGER":
        GOTO 10
    30 FOR I = 1 TO N
    40     IF N/I = INT(N/I) THEN PRINT I;
    50 NEXT I

22A. 10 FOR I = 1 TO 20
    20     R = RND(101) - 1
    30     PRINT R;
    40 NEXT I

22B. 10 FOR I = 1 TO 12
    20     R = RND(25) - 1
    30     PRINT R;
    40 NEXT I
    50 FOR I = 1 TO 8
    60     R = RND(76) + 24
    70     PRINT R;
    80 NEXT I

24. 10 SET (64,17): REM THIS IS THE TOP POINT
    20 FOR I = 18 TO 30
    30     SET((81-I),I): REM THIS DRAWS THE LEFT SIDE
    40     SET((47+I),I): REM THIS DRAWS THE RIGHT SIDE
    50 NEXT I
    55 REM THIS IS THE FINAL UNDERLINE
    60 FOR J = 1 TO 29
    70     SET((49+J),31)
    80 NEXT J

```

```

26A. 5 CLS
10 FOR X = 3 TO 125
20     SET(X,3) : SET(X,45): REM DRAWS OUTSIDE HORIZONTALS
30 NEXT X
40 FOR X = 55 TO 73
50     SET(X,18) : SET(X,30): REM DRAWS INSIDE HORIZONTALS
60 NEXT X
70 FOR Y = 18 TO 30
80     SET(55,Y) : SET(73,Y): REM DRAWS INSIDE VERTICALS
90 NEXT Y
100 FOR Y = 3 TO 45
110     SET(3,Y) : SET(125,Y): REM DRAWS OUTSIDE VERTICALS
120 NEXT Y
130 FOR I = 1 TO 10
140     X1 = RND(128)-1 : Y1 = RND(48)-1
150     SET(X1,Y1)
160 NEXT I

```

```

26B. 5 CLS
10 FOR X = 3 TO 125
20     SET(X,3) : SET(X,45): REM DRAWS OUTSIDE HORIZONTALS
30 NEXT X
40 FOR X = 55 TO 73
50     SET(X,18) : SET(X,30): REM DRAWS INSIDE HORIZONTALS
60 NEXT X
70 FOR Y = 18 TO 30
80     SET(55,Y) : SET(73,Y): REM DRAWS INSIDE VERTICALS
90 NEXT Y
100 FOR Y = 3 TO 45
110     SET(3,Y) : SET(125,Y): REM DRAWS OUTSIDE VERTICALS
120 NEXT Y
130 FOR I = 1 TO 10
140     X1 = RND(128)-1 : Y1 = RND(48)-1
150     SET(X1,Y1)
155 IF X1 > 54 AND X1 < 74 AND Y1 > 17 AND Y1 < 31
        THEN S = S + 10
        ELSE IF X1 > 2 AND X1 < 126 AND Y1 > 2 AND Y1 < 46
            THEN S = S + 4
            ELSE S = S - 1
160 NEXT I
165 FOR I = 1 TO 1000 : NEXT I : REM THIS IS A DELAY
167 PRINT@ 960," "
170 PRINT@ 960,"SCORE IS";S;

```

```

28.  10 CLS
    20 FOR X = 0 TO 127 : SET(X,41) : NEXT X : REM BORDER
    30 X1 = RND(128)-1 : Y1 = RND(40)-1
    40 X2 = RND(128)-1 : Y2 = RND(40)-1
    50 SET(X1,Y1) : SET(X2,Y2)
    60 FOR I = 1 TO 10
    70     PRINT@ 896,"COORDINATE";:INPUT X,Y
    80     IF POINT(X,Y) = -1 THEN RESET(X,Y):H = H + 1 :
        PRINT@ 896,"A HIT!"
    90 SET(X,Y) :FOR J = 1 TO 500: NEXT J :RESET(X,Y): REM SHOW SHOT
    100 IF H = 2 THEN 140
    110 PRINT@ 896,"" : REM CLEAR OLD GUESS
    120 NEXT I
    130 PRINT "YOU FAILED TO GET THEM YOU TURKEY!" : END
    140 PRINT "BRILLIANT WORK YOU GOT THEM BOTH!"

30.  10 SET(15,18)
    20 FOR Y = 20 TO 24
    30     SET(15,Y)
    40 NEXT Y

32.  10 CLS
    20 INPUT "A,B,R";A,B,R
    30 FOR X = A - R TO A + R
    40     Y1 = INT(B + SQR(R[2 - (X-A)[2] + .5)
    50     Y2 = INT(B - SQR(R[2 - (X-A)[2] + .5)
    60     SET(X,Y1) : SET (X,Y2)
    70 NEXT X

```



```

2.      >RUN
        .
        ...
        .....
        .....
        .....
READY
>
>50
>RUN
.....

4.      5 DIM L$(15)
        10 FOR I = 1 TO 15
        20     READ L$(I)
        30 NEXT I
        40 FOR I = 15 TO 1 STEP -1
        50     PRINT L$(I);
        60 NEXT I
        100 DATA P,U,O,R,G,S,D,R,A,W,K,C,A,B,A

6.      10 DIM A$(11,3)
        20 FOR I = 1 TO 11
        30     FOR J = 1 TO 3
        40         READ A$(I,J)
        50         PRINT A$(I,J);
        60     NEXT J
        70     PRINT " ";
        80 NEXT I
        90 PRINT:PRINT
        100 FOR J = 1 TO 3
        110     FOR I = 1 TO 11
        120         PRINT A$(I,J);
        130     NEXT I
        135     PRINT " ";
        140 NEXT J
        150 PRINT
        200 DATA A,L,O,T,O,F,T,R,O,U,A
        210 DATA L,E,W,H,A,T,I,S,T,H,E
        220 DATA M,A,T,T,E,R,W,I,T,H,M

8.      10 FOR A = 3 TO 5
        20     FOR B = 1 TO 4
        30         IF A * B <= 10 THEN 50
        40         PRINT A,B
        50     NEXT B
        60 NEXT A

10.     >RUN
        B4= 1   BUT B(4)= 86
        | 14           125           111

```



```

12.  10 FOR X = 1 TO 5
      20   READ N$(X), A$(X), T$(X), S$(X), Z$(X)
      30 NEXT X
      40 INPUT "NAME"; N$
      50 FOR X = 1 TO 5
      60   IF N$(X) = N$ THEN 100
      70 NEXT X
      80 PRINT "PERSON NOT FOUND."
      90 GOTO 40
     100 PRINT N$(X)
     110 PRINT A$(X)
     120 PRINT T$(X); ", "; S$(X); " "; Z$(X)
     130 GOTO 40
     140 DATA DON MIKAN, 42 PENNSYLVANIA AVE., WASHINGTON, DC, 11114
     150 DATA LESTER WATERS, PORTA NIGRA, WEST BERLIN, GN, 00000
     160 DATA ROB LYNCH, DRUMTHWACKETT, PRINCETON, NEW JERSEY, 08540
     170 DATA ELI HUROWITZ, SMALL ST., TERRE HAUTE, INDIANA, 32226
     180 DATA BRUCE PRESLEY, OLD LOWER, WESTHAMPTON, NY, 01234

```

```

14.  10 DIM I$(3,5)
      20 FOR A = 1 TO 3
      30   FOR I = 1 TO 5
      40     READ I$(A,I)
      50   NEXT I
      60 NEXT A
      70 INPUT "WHAT ARE YOU LOOKING FOR "; A$
      80 FOR A = 1 TO 3
      90   FOR I = 1 TO 5
     100     IF I$(A,I) = A$ THEN 220
     110   NEXT I
     120 NEXT A
     200 PRINT "I'M SORRY, WE DON'T HAVE "; A$
     210 GOTO 70
     220 PRINT "YOU WILL FIND "; A$; " IN AISLE #"; A; " ITEM #"; I
     230 GOTO 70
     300 DATA PEANUTS, KETCHUP, SALT, CHERRIES, CEREAL
     310 DATA BREAD, BUTTER, MILK, SUGAR, CHEESE
     320 DATA BEEF, CHICKEN, LAMB, HAM, PORK

```

```

16.  10 FOR X = 1 TO 5
      20   PRINT "STUDENT #"; X;
      30   INPUT N$(X)
      40   INPUT "ENTER FOUR GRADES: "; G1(X), G2(X), G3(X), G4(X)
      50   A(X) = (G1(X) + G2(X) + G3(X) + G4(X))/4
      60 NEXT X
      70 PRINT "NAME"; TAB(15); "1"; TAB(20); "2";
      80 PRINT TAB(25); "3"; TAB(30); "4"; TAB(36); "AVE."
      90 FOR X = 1 TO 5
     100   PRINT N$(X); TAB(13); G1(X); TAB(18); G2(X);
     110   PRINT TAB(23); G3(X); TAB(28); G4(X); TAB(34); A(X)
     120   A = A + A(X) : REM   RUNNING SUM OF STUDENTS AVERAGES
     130 NEXT X
     140 PRINT TAB(35); "-----"
     150 PRINT TAB(34); A/5

```

```

18.  10 FOR I = 1 TO 6
      20 READ T$(I) : REM POSSIBLE TOPS
      30 NEXT I
      40 FOR I = 1 TO 5
      50 READ B$(I) : REM POSSIBLE BOTTOMS
      60 NEXT I
      70 FOR T = 1 TO 6
      80 FOR B = 1 TO 5
      90 PRINT T$(T);" WITH ";B$(B)
     100 NEXT B
     110 NEXT T
     200 DATA BLUE SHIRT,PINK SHIRT,ORANGE SHIRT,BLACK SHIRT
     210 DATA YELLOW SWEATER,GREEN SWEATER
     220 DATA RED SKIRT,GREEN SKIRT,PURPLE SKIRT
     230 DATA WHITE PANTS,ELECTRIC PURPLE PANTS

```

20a. >RUN

```

0
1
2
2
3
4
5
7
8
9
READY
>

```

b.

```

>RUN
SUSAN
ROB
MARY
LIZ
KIM
ELI
FAZIOLI
LESTER
FRENCH
DON
READY
>

```

```

22.  1 REM ARRAY P HOLDS NUMBER OF TIMES EACH ROLL OCCURS
      2 REM D1 = DIE #1
      3 REM D2 = DIE #2
      10 DIM P(12)
      20 FOR I = 1 TO 1000
      30 D1 = RND(6)
      40 D2 = RND(6)
      50 P(D1 + D2) = P(D1 + D2) + 1
      60 NEXT I
      70 PRINT "POINT TOTAL","TIMES APPEARING"
      80 FOR I = 2 TO 12
      90 PRINT I,P(I)
     100 NEXT I

```

```

24.  10 DIM A$(8,8)
    20 REM   SET UP THE RANDOM BOARD
    30 FOR R = 1 TO 8
    40     FOR C = 1 TO 8
    50         IF RND(2) = 1
                THEN A$(R,C) = "X"
                ELSE A$(R,C) = "O"
    60     NEXT C
    70 NEXT R
    80 REM   DISPLAY BOARD
    90 CLS
    100 PRINT TAB(11); "1 2 3 4 5 6 7 8"
    110 FOR R = 1 TO 8
    120     PRINT TAB(7); R;
    130     FOR C = 1 TO 8
    140         PRINT " "; A$(R,C);
    150     NEXT C
    160     PRINT
    170 NEXT R
    180 INPUT "ROW, COLUMN"; R,C
    190 REM   FIND OUT WHAT IS AT A$(R,C) AND SET C$ TO ITS REVERSE
    200 T$ = A$(R,C)
    210 IF T$ = "X"
            THEN C$ = "O"
            ELSE C$ = "X"
    220 REM   NOW SET EVERYTHING IN COLUMN C TO C$
    230 FOR V = 1 TO 8
    240     A$(V,C) = C$
    250 NEXT V
    260 REM   NOW SET EVERYTHING IN ROW R TO C$
    270 FOR K = 1 TO 8
    280     A$(R,K) = C$
    290 NEXT K
    300 GOTO 80

26.  5 CLS
    10 DIM N(10)
    20 FOR T = 1 TO 200
    30     X = RND(10) : REM GET NUMBERS
    40     IF N(X) < 40 THEN N(X) = N(X) + 1 : REM MAKE SURE NO OVFL
    50 NEXT T
    60 REM
    100 REM DO BAR CHART
    110 FOR I = 1 TO 10
    120     FOR P = N(I) TO 40
    130         IF P = 40 THEN 150 : REM DON'T PLOT ON 40
    140         SET(I * 10, P) : SET(I * 10 + 1, P)
    150     NEXT P
    160 NEXT I
    170 REM
    200 REM DRAW BASE
    205 FOR X = 0 TO 127
    210     SET(X, 40)
    220 NEXT X
    230 REM LABEL GRAPH
    240 FOR I = 10 TO 1 STEP -1
    250     PRINT@ 896, TAB(5*I-1); I
    260 NEXT I

```

2.

```

RUN
<10><20><30><40><50><60><40><50><60><40><50><60><70> 20 44
<80><30><40><50><60><40><50><60><40><50><60><70> 1364 2732
<80><30><40><50><60><40><50><60><40><50><60><70> 87380 174764
<80><90><100>

```

4.

a) In its current state, this program will continuously print the sum of one, two, three. Line 30 should be changed as follows:

```
30 GOTO 10
```

This allows new data to be read.

b) This program will print "IS LESS THAN 5" when X is in fact greater than five. Line 20 may be corrected as follows:

```
20 IF X<=5 THEN 30
```

Line 50 should be changed to read:

```
50 PRINT X;"IS LESS THAN OR EQUAL TO FIVE"
```

c) Line 40 of this program has no effect. Either way, program control is transferred to line 50. The line may be corrected as follows:

```
40 IF Y>1 THEN 60
```

6.

The problem lies in the routine that attempts to switch the two values referenced by the variables "A" and "B" (lines 30 and 40). At line 30, "A" assumes the value of "B", and the previous contents of "A" are lost. In line 40, "B" takes on the value of "A", so that both "A" and "B" now contain the original value of "B". To solve this problem a dummy variable such as "D" must be introduced.

```
30 D=A : A=B
40 B=D
```

Thus the original value of "A" is not lost.

8.

```
a) 10 REM    THIS PROGRAM DECIDES WHICH MOVIE WE WILL SEE
20 RANDOM
30 R=RND(20)+5
40 X=1
50 FOR L=1 TO R
60     X=X*-1
70     IF L=R
        THEN PRINT "WE WILL SEE ";
        ELSE PRINT
80     IF X>0
        THEN PRINT "ANNIE"
        ELSE PRINT "POLTERGEIST"
90 NEXT L
100 END
```

```
b) 10 DIM J(10,10)
20 FOR X=0 TO 9
30     PRINT TAB(X*4+5);X+1;
40 NEXT X
50 PRINT
60 FOR X=0 TO 8
70     PRINT TAB(X*4+6);"--";
80 NEXT X
90 PRINT TAB(42);"--"
100 FOR X=1 TO 10
110     FOR Y=1 TO 10
120         J(X,Y)=X+Y
130         IF F=0 THEN PRINT X;":";
140         PRINT TAB(F*4+5);J(X,Y);
150         F=F+1
160         IF F=10 THEN PRINT: F=0
170     NEXT Y
180 NEXT X
190 END
READY
>
```

CHAPTER 6

2. 10 PRINT "NUMBER","SQUARE ROOT"
 20 FOR X = 121 TO 144
 30 PRINT X,SQR(X)
 40 NEXT X

4. 10 INPUT N
 20 IF N = 0 THEN PRINT "0" : END
 30 PRINT ABS(N)/N

6. 10 INPUT N
 20 PRINT SQR(N[2])

8. 10 INPUT A
 20 A1 = A/360
 30 A2 = A * 3.1415926535/180
 40 PRINT A;"DEGREES=";A1;"REVS=";A2;"RADIANS"

10. 10 PRINT "ANGLE";TAB(7);"SINE";TAB(19);"COSINE";TAB(28);
 "SUM OF SQUARES"
 20 FOR X = 0 TO 180 STEP 10
 30 A = X * 3.141592653/180
 40 PRINT X;TAB(6);SIN(A);TAB(19);COS(A);TAB(33);SIN(A)[2+
 COS(A)[2
 50 NEXT X

12. 10 DEF FNF(X) = 20 * SQR(ABS(X)) - 10 * SGN(X) + 5*INT(X)
 20 INPUT "WHAT IS THE NUMBER";N
 30 PRINT "F(N) = ";FNF(N)
 40 PRINT "F(F(N)) = ";FNF(FNF(N))

14. 10 PRINT " X";TAB(7);"LOG(X)";TAB(24);"10[X"
 20 FOR X = 1 TO 15
 30 PRINT X;TAB(5);LOG(X)/LOG(10);TAB(22);10[X
 40 NEXT X

16. 10 FOR X = 50 TO 60
 20 PRINT SQR(X)
 30 NEXT X

18. 10 INPUT N
 20 PRINT "SQR(";N;") ="; SQR(N)
 30 PRINT "SGN(";N;") ="; SGN(N)
 40 PRINT "LOG(";N;") ="; LOG(N)
 50 PRINT "SIN(";N;") = "; SIN(N)

```

20. 10 DEF FNA(X) = X*X + 3*X +2
    20 DEF FNB(X) = LOG(X*X+1) - X
    30 DEF FNC(X) = ATN(SIN(X))
    40 FOR X = -10 TO 10
    50     PRINT FNA(X),FNB(X),FNC(X)
    60 NEXT X

22. a. 10 PRINT " X";TAB(7);"EXP(X)";TAB(25);"LOG(EXP(X))"
    20 FOR X = -5 TO 10
    30     PRINT X;TAB(5);EXP(X);TAB(28);LOG(EXP(X))
    40 NEXT X

    b. 10 PRINT " X";TAB(7);"LOG(X)";TAB(25);"EXP(LOG(X))"
    20 FOR X = 1 TO 151 STEP 10
    30     PRINT X;TAB(5);LOG(X);TAB(26);EXP(LOG(X))
    40 NEXT X

24. 10 P = 0.50 : REM  INITIAL INVESTMENT
    20 P = P * EXP(.08 * 10) : REM  FIRST TEN YEARS
    30 FOR X = 11 TO 50 : REM  THE 11'TH TO 50'TH YEAR
    40     P = P - .05 : REM  THE NICKEL WITHDRAWN
    50     P = P * EXP(.08 * 1)
    60 NEXT X
    70 PRINT "AFTER 50 YEARS DENNIS HAS $";P

26A. 10 PI = 3.1415926535
    20 DEF FNS(C) = ATN(C/SQR(1 - C*C)) : REM  ARCSINE
    30 INPUT "SIDES 1 AND 2";S1,S2
    40 INPUT "INCLUDED ANGLE";A3
    50 A3 = A3 * PI/180 : REM  CONVERT TO RADIANS
    60 S3 = SQR(S1*S1 +S2*S2 -2*S1*S2*COS(A3)) :REM  LAW OF COSINES
    70 K = SIN(A3) / S3 : REM  CONSTANT FOR USE OF LAW OF SINES
    80 A2 = FNS(K * S2) : REM  LAW OF SINES
    90 A1 = FNS(K * S1)
    100 A1 = A1 * 180 / PI : REM  CONVERT TO DEGREES
    110 A2 = A2 * 180 / PI
    120 A3 = A3 * 180 / PI
    130 PRINT "SIDE","ANGLE"
    140 PRINT S1,A1
    150 PRINT S2,A2
    160 PRINT S3,A3

```

```

26B. 10 PI = 3.1415926535
      20 INPUT "ANGLES 1 AND 2";A1,A2
      30 INPUT "SIDE";S
      40 INPUT "IS THAT THE INCLUDED SIDE";A$
      50 IF A$ = "Y" THEN S3 = S : GOTO 100
      60 S1 = S : REM NAME THE SIDES PROPERLY
      70 INPUT "ADJACENT TO WHICH ANGLE";S0
      80 REM SET UP THE ANGLES FOR PROPER CALCULATIONS
      90 IF S0 = 1 THEN S0 = A1 : A1 = A2 : A2 = S0
      100 A3 = 180 - A1 - A2 : REM GET THIRD ANGLE
      110 A1 = A1 * PI / 180 : REM CONVERT TO RADIANS
      120 A2 = A2 * PI / 180
      130 A3 = A3 * PI / 180
      140 REM USE LAW OF SINES FOR CALCULATION OF S2 AND S3
      150 IF S3 = 0 THEN K = S1 / SIN(A1)
           :S2 = K * SIN(A2)
           :S3 = K * SIN(A3)
           :GOTO 180
      160 K = SIN(A3) / S3
      170 S1 = SIN(A1) / K : S2 = SIN(A2) / K
      180 A1 = A1 * 180 / PI
      190 A2 = A2 * 180 / PI
      200 A3 = A3 * 180 / PI
      210 PRINT "SIDE","ANGLE"
      220 PRINT S1,A1
      230 PRINT S2,A2
      240 PRINT S3,A3

```

```

26C. 10 PI = 3.1415926535
      20 DEF FNC(X) = ATN(SQR(1 - X*X) / X) : REM ARCCOSINE
      30 INPUT "3 SIDES";S1,S2,S3
      40 REM USE LAW OF COSINES TO DEDUCE ANGLES
      50 A1 = (S2*S2 + S3*S3 - S1*S1) / 2 / S2 / S3
      60 A2 = (S1*S1 + S3*S3 - S2*S2) / 2 / S1 / S3
      70 A3 = (S1*S1 + S2*S2 - S3*S3) / 2 / S1 / S3
      75 REM MAKE SURE AN ANGLE'S COSINE IS NOT 0
           IF IT IS SET THE ANGLE TO PI/2
      80 IF A1 = 0
           THEN A1 = PI/2
           ELSE A1 = FNC(A1)
      90 IF A2 = 0
           THEN A2 = PI / 2
           ELSE A2 = FNC(A2)
      100 IF A3 = 0
           THEN A3 = PI / 2
           ELSE A3 = FNC(A3)
      110 REM CHECK IF ANGLES ARE LESS THAN ZERO AND ADJUST THEM
      120 IF A1 < 0 THEN A1 = A1 + PI
      130 IF A2 < 0 THEN A2 = A2 + PI
      140 IF A3 < 0 THEN A3 = A3 + PI
      150 REM NOW CONVERT TO DEGREES
      160 A1 = A1 * 180 / PI
      170 A2 = A2 * 180 / PI
      180 A3 = A3 * 180 / PI
      190 PRINT "SIDE","ANGLE"
      200 PRINT S1,A1
      210 PRINT S2,A2
      220 PRINT S3,A3

```



```

28A. 10 REM    XO,YO = Origin of graph on screen
20 REM    X,Y    = Function variables
30 REM    X1,Y1 = Shifted/Scaled X & Y values
40 REM    T      = Loop variable for setting up axes
50 REM
60 XO = 64 : YO = 25 : REM Origin (0,0)
70 REM
80 REM Draw and label X-axis
90 CLS : FOR T = 0 TO 127 : SET(T,YO) : NEXT T
100 FOR T = 4 TO 124 STEP 20
110     SET(T,YO-1) : SET(T,YO+1)
120     READ N : REM Read a label from DATA statement
130     IF N > 0
140         THEN PRINT@ (574 + INT(T/2)), N;
150         ELSE PRINT@ (446 + INT(T/2)), N;
160 NEXT T
170 REM
180 REM Draw and label Y-axis
190 FOR T = 3 TO 47 : SET(XO,T) : NEXT T
200 FOR T = 5 TO 45 STEP 10
210     SET(XO-1,T) : SET(XO+1,T)
220 NEXT T
230 PRINT @94,"1"; @993,"-1"; @349,".5";
240     @737,"-.5"; @29,"Y-axis";
250 REM
260 REM Plot function Y = 1/X
270 FOR X = -12 TO 12 STEP 0.25
280     Y = SGN(X)
290     REM
300     REM Shift/Scale X & Y to display screen
310     X1 = XO + X * (120/24)
320     Y1 = YO - Y * (40/2)
330     IF (X1 > 0 AND Y1 > 0)
340         THEN IF (X1 < 128 AND Y1 < 48)
350             THEN SET(X1,Y1)
360 NEXT X
370 DATA -12, -8, -4, 0, 4, 8, 12

```

For parts B and C change the following lines.

B. 250 Y = ATN(X)

C. 210 PRINT @94,"12"; @993,"-12"; @349," 6";
@737,"-6"; @29,"Y-axis";
250 Y = INT(X)
290 Y1 = YO - Y * (25/12)

2.

```

10 A$="YOU HAVE WON A "
20 R=RND(4)
30 ON R GOTO 40,60,80,100
40 PRINT A$;"BALL"
50 END
60 PRINT A$;"BALLOON"
70 END
80 PRINT A$;"TOY CAR"
90 END
100 PRINT A$;"CANDY BAR"
110 END

```

4.

```

READY
>RUN
ROW, ROW, ROW, YOUR BOAT
GENTLY DOWN THE STREAM
MERRILY, MERRILY, MERRILY, MERRILY,
LIFE IS BUT A DREAM
READY
>

```

6a.

```

>RUN
<10><20><30><60><70><80><40><50><10><20><30><60><70><80><40><50>
<10><20><30><60><80><40><50><10><20><95><100>
READY

```

6b.

```

>RUN
T= 2          F= 20
READY

```

```

8. 10 X=RND(11)-6 : Y=RND(11)-6
    20 FOR I=1 TO 3
    30     INPUT "TWO INTEGERS BETWEEN -5 AND 5, INCLUSIVE";X1,Y1
    40     GOSUB 140
    50     IF D=0 THEN END
    60 NEXT I
    70 X1=X: PRINT "X=";X
    80 FOR J=1 TO 2
    90     INPUT "Y1";Y1
    100    GOSUB 140
    110    IF D=0 THEN END
    120 NEXT J
    130 PRINT "YOU FAILED. Y=";Y
    135 END
    140 D=SQR((X-X1)2+(Y-Y1)2)
    150 PRINT D
    160 IF D=0 THEN PRINT "CORRECT"
    170 RETURN

```

```

10. 10 READ A,B,C
    20 GOSUB 100
    30 PRINT "AREA ="A1
    40 GOTO 10
    50 REM
    100 REM    SUBROUTINE TO CALCULATE AREA
    110 S = (A + B + C)/2
    120 A1 = SQR(S * (S - A) * (S - B) * (S - C))
    130 RETURN
    140 REM
    200 DATA 5,8,12
    210 DATA 8,13,20
    220 DATA 3,4,5

```

```

12. 10 ON ERROR GOTO 100
    20 INPUT "A,B,C";A,B,C
    30 D=SQR(B*B-4*A*C)
    40 R1=(-B+D)/2*A : R2=(-B-D)/2*A
    50 PRINT "THE ROOTS ARE ";R1;"AND ";R2
    60 GOTO 20
    100 IF ERR/2+1=5
        THEN PRINT "IMAGINARY ROOTS ENCOUNTERED":
            RESUME 20
    110 PRINT "UNEXPECTED ERROR ENCOUNTERED"
    115 RESUME 20
    120 END

```

2.

```

1 REM   A$,B$,C$ = INPUT STRINGS
2 REM   S = SUM OF ASC'S
3 REM
4 REM
10 INPUT "THREE SEPARATE LETTERS";A$,B$,C$
20 S=ASC(A$)+ASC(B$)+ASC(C$)
30 I=INT(S/3) : REM   AVERAGES THE ASCII VALUES
40 PRINT CHR$(I)
50 END

```

4.

```

1 REM   A$,B$ = ASCII CHARACTERS OF 45 AND 46 RESPECTIVELY
2 REM   X,Y,Z = LOOP DUMMY VARIABLES
3 REM
4 REM
10 A$=CHR$(45)
20 B$=CHR$(46)
30 PRINT B$;
40 FOR X=1 TO 8
50     PRINT A$;
60 NEXT X
70 PRINT B$
80 FOR Y=1 TO 3
90     PRINT B$;
100    PRINT TAB(9);B$
110 NEXT Y
120 PRINT B$;
130 FOR Z=1 TO 8
140     PRINT A$;
150 NEXT Z
160 PRINT B$
170 END

```

6.

```

1 REM   L = LENGTH OF THE STRING
2 REM   F = ASC OF THE FIRST CHARACTER
3 REM   T = ASC OF THE LAST CHARACTER
4 REM
5 REM
10 INPUT "YOUR STRING? ";A$
20 L=LEN(A$)
30 F=ASC(A$)
40 T=ASC(RIGHT$(A$,1))
50 PRINT "IT HAS ";L;" CHARACTERS"
60 PRINT "IT STARTS WITH ASCII NUMBER ";F
70 PRINT "IT ENDS WITH ASCII NUMBER ";T
80 END

```

8.

```
10 A$="2598"
20 PRINT VAL(LEFT$(A$,2))+VAL(RIGHT$(A$,2))
30 END
```

10. Decimal	Binary
89	1011001
74	1001010
80	1010000
107	1101011
255	11111111
129	10000001
28	11100
39	100111
29	11101
24	11000
43	101011

```
12. 1 REM   A$ = A 10 CHARACTER STRING OF RANDOM DIGITS
    2 REM   N1 = VALUE OF THE FIRST THREE DIGITS
    3 REM   N2 = VALUE OF THE LAST THREE DIGITS
    4 REM
    5 REM
    10 INPUT "ENTER THE DIGITS 0-9 IN RANDOM ORDER: ";A$
    20 IF LEN(A$)<>10 THEN 10
    30 N1=VAL(LEFT$(A$,3))
    40 N2=VAL(RIGHT$(A$,3))
    50 PRINT "THE SUM OF";N1;"AND";N2;"IS";N1+N2
    60 END
```

```
14. 1 REM   N$ = THE ORIGINAL NAME ENTERED
    2 REM   B$ = A STRING TO BUILD THE NAME BACKWARDS
    3 REM   P% = POSITION IN THE ORIGINAL NAME
    4 REM
    5 REM
    9 CLEAR 100
    10 INPUT "WHAT IS THE NAME";N$
    20 FOR P%=LEN(N$) TO 1 STEP -1
    30     B$=B$+MID$(N$,P%,1)
    40 NEXT P%
    50 PRINT """;N$;"' BACKWARDS IS "'";B$;"'"
    60 END
```

16. READY
 >RUN
 FOXY
 SAYS
 HELP
 WILL
 COME
 SOON
 READY
 >

18. 1 REM L% = LENGTH OF A WORD
 2 REM W% = A STRING TO BUILD A WORD A CHARACTER AT A TIME
 3 REM
 4 REM
 10 FOR X%=1 TO 20
 20 W\$=""
 30 FOR L%=1 TO RND(4%)+1%
 40 W\$=W\$+CHR\$(RND(26%)+64%)
 50 NEXT L%
 60 PRINT W\$;" ";
 70 NEXT X%
 80 END

20. 1 REM W\$ = SECRET WORD
 2 REM G\$ = WORD GUESSED
 3 REM A\$,B\$ = LETTERS IN WORDS W AND G RESPECTIVELY
 4 REM
 5 REM
 10 W\$="HANGMAN"
 20 INPUT "GUESS A WORD"; G\$
 30 IF G\$=W\$ THEN PRINT "YOU GUESSED IT" : GOTO 120
 40 FOR I%=1 TO LEN(G\$)
 50 B\$=MID\$(G\$,I%,1)
 60 FOR J%=1 TO LEN(W\$)
 70 A\$=MID\$(W\$,J%,1)
 80 IF B\$=A\$ THEN PRINT B\$;" IS IN THE WORD":GOTO 100

 90 NEXT J%
 100 NEXT I%
 110 GOTO 20
 120 END

22. Since the computer works with base two, it expresses any power of two, or sum of powers of two exactly. However, a number such as $1/3$ cannot be represented as a sum of powers of two. In addition, the fraction $1/3$ does not have an exact decimal representation. In the case of $1/2$, the computer can represent the fraction more accurately.

24.

```

1 REM   T$ = A WORD FROM THE SENTENCE
2 REM   W$( ) = STORES EACH WORD (UP TO 30)
3 REM   S$ = THE ORIGINAL SENTENCE
4 REM
5 REM
10 CLEAR 250
20 DIM W$(30)
30 INPUT "YOUR SENTENCE"; S$
40 S$=CHR$(32)+S$+CHR$(32)
50 FOR X%=1 TO LEN(S$)
60     T$=MID$(S$,X%,1)
70     IF ASC(T$)=32 THEN C%=C%+1 : GOTO 90
80     W$(C%)=W$(C%)+T$
90 NEXT X%
100 FOR L%=30 TO 1 STEP -1
110     IF W$(L%)<>"" THEN PRINT W$(L%); " ";
120 NEXT L%
130 END

```

26.

```

1 REM   A = HEIGHT
2 REM   B = DISTANCE TO INDENT THE BOTTOM
3 REM   C = WHERE BOTTOM EDGE ENDS
4 REM
5 REM
10 CLEAR 200
20 S$="TRIANGLE"
30 INPUT "A,B,AND C";A,B,C
40 S$=S$+S$
50 IF LEN(S$)< (C+LEN("TRIANGLE")) THEN 40
60 FOR H%=1 TO A
70     P1%=(B/A)*H%
80     P2%=(C/A)*H%
90     IF P1%<1 THEN P1%=P1%+1
100    PRINT TAB(P1%); MID$(S$,P1%,ABS(P2%-P1%))
110 NEXT H%
120 END

```

- 2A. 10 FOR X = 1 TO 5
 20 INPUT "DESSERT ";N\$(X)
 30 INPUT "PRICE";P(X)
 40 NEXT X
 50 REM
 100 REM NOW PRINT TO THE FILE
 110 OPEN "O",1,"NAPRI:1"
 120 FOR I = 1 TO 5
 130 PRINT #1,N\$(I);", ";P(I)
 140 NEXT I
 150 CLOSE 1
- 2B. 10 FOR X = 1 TO 2
 20 INPUT "DESSERT ";N\$(X)
 30 INPUT "PRICE";P(X)
 40 NEXT X
 50 REM
 100 REM NOW PRINT TO THE FILE
 110 OPEN "E",1,"NAPRI:1"
 120 FOR I = 1 TO 2
 130 PRINT #1,N\$(I);", ";P(I)
 140 NEXT I
 150 CLOSE 1
- 2C. 10 OPEN "I",1,"NAPRI:1"
 20 IF EOF(1) THEN 60
 30 INPUT #1,N\$,P
 40 PRINT N\$,P
 50 GOTO 20
 60 CLOSE 1
- 4A. 1 REM N\$ = ARRAY TO HOLD THE TEN NAMES
 10 REM GET THE NAMES
 20 FOR X = 1 TO 10
 30 INPUT "NAME ";N\$(X)
 40 NEXT X
 50 REM
 100 REM NOW PRINT THEM TO THE FILE TNM
 110 OPEN "O",1,"TNM:1"
 120 FOR I = 1 TO 10
 130 PRINT #1,N\$(I)
 140 NEXT I
 150 CLOSE 1
- 4B. 1 REM X = A FLAG TO SEE IF A NAME WAS FOUND
 2 REM N\$ = THE NAME FOUND IN THE FILE TNM
 10 X = -1 : REM WILL BE SET > 0 IF A NAME IS FOUND IN RANGE
 20 OPEN "I",1,"TNM:1"
 30 FOR I = 1 TO 10
 40 INPUT #1,N\$
 50 IF N\$ > "C" AND N\$ < "I" THEN PRINT N\$: X = X + 2
 60 NEXT I
 70 CLOSE 1
 80 IF X < 0 THEN PRINT "NONE FOUND"

6A.

```

1 REM   N$ = NAME
2 REM   R = HOURLY PAY RATE
3 REM   D = NUMBER OF DEPENDENTS
4 REM   M = MEDICAL INSURANCE DEDUCTION
5 REM   L = LIFE INSURANCE DEDUCTION
10 OPEN "O",1,"PAY:1"
20 FOR X = 1 TO 10
30     READ N$,R,D,M,L
40     PRINT #1,N$;" ";R;" ";D;" ";M;" ";L
50 NEXT X
60 CLOSE 1
70 REM
100 DATA ISLAND GILLIGAN,-4.50,0,2,1
110 DATA WILSON MISTER,4.95,2,5,3
120 DATA MENACE DENNIS,1,0,2,1
130 DATA BEAVER LEAVEITTO,2.25,0,1,1
140 DATA HOGAN COLONEL,40,2,3,3
150 DATA SCHULTZ SEARGENT,30,1,1,2
160 DATA KIRK JIM,78.50,1,1,1
170 DATA BUNNY BUGS,12.45,2,5,3
180 DATA SERLING ROD,55.90,2,2,1
190 DATA TARZAN ME,9.25,3,0,0

```

6B.

```

1 REM   N$ = NAME
2 REM   R = HOURLY PAY RATE
3 REM   D = NUMBER OF DEPENDENTS
4 REM   M = MEDICAL INSURANCE DEDUCTION
5 REM   L = LIFE INSURANCE DEDUCTION
10 DEF FNR(X) = INT(100 * X + .5)/100 : REM ROUND TO 100TH
20 OPEN "I",1,"PAY:1"
30 FOR X = 1 TO 10
40     INPUT #1,N$,R,D,M,L
50     READ H : REM GET HOURS WORKED
60     G = H * R : T = (25 - 2*D) * G / 100
70     IF T < 0 THEN T = 0
80     P = G - T - M - L
90 PRINT N$ : FOR L = 1 TO LEN(N$) : PRINT "-"; : NEXT L : PRINT
91 PRINT "GROSS PAY:";TAB(40);FNR(G);PRINT "TAX:";
92 PRINT TAB(40);FNR(T) : PRINT "MEDICAL INSURANCE:";TAB(40);
93 PRINT FNR(M);PRINT "LIFE INSURANCE:";TAB(40);FNR(L)
100 PRINT "NET:";TAB(40);FNR(P)
105 PRINT
110 NEXT X
120 CLOSE 1
130 REM
200 DATA 40,37.1,51,48,39.2,18.6,59.8,45.3,68.3,41

```

```

8A.  10 OPEN "O",1,"SEATS:1"
      20 FOR X = 1 TO 20
      30   PRINT #1," ",0"
      40 NEXT X
      50 CLOSE 1

8B.  10 DIM N$(20),P(20)
      20 OPEN "O",1,"SEATS:1"
      30 FOR X = 1 TO 20
      40   READ N$(X)
      50 NEXT X
      60 REM
      100 REM   NOW PRINT THE STUDENTS RANDOMLY
      110 FOR I = 1 TO 20
      120   S = RND(20)
      130   IF P(S) = 1 THEN 120
      140   P(S) = 1
      150   PRINT #1,N$(S);", ";I
      160 NEXT I
      170 CLOSE 1
      200 DATA SPANKY,BUCKWHEAT,ALFALFA,PETHEY,BUTCH,DARLA
      210 DATA LAUREL,HARDY,LARRY,CURLY,MOE,ABBOTT,COSTELLO
      220 DATA FRED,WILMA,BARNEY,BETTY,PEBBLES,BAM-BAM,DINO

8C.  1 REM   SN = SEAT NUMBER
      2 REM   N$ = NAME
      10 OPEN "I",1,"SEATS:1"
      15 PRINT "NAME","SEAT"
      20 FOR I = 1 TO 20
      30   INPUT #1,N$,SN
      40   PRINT N$,SN
      50 NEXT I
      60 CLOSE 1

```


CHAPTER 10

```

2A.  1 REM   T$,T1$ = TITLE (30 CHARACTERS)
      2 REM   T = BOOK NUMBER
      3 REM
      4 REM
      10 CLEAR 75
      20 OPEN "R",1,"BOOKS/TXT:1",30
      30 FIELD 1, 30 AS T$
      40 FOR T = 1 TO 10
      50     INPUT "TITLE";T1$ : LSET T$ = T1$
      60     PUT 1
      70 NEXT T
      80 CLOSE 1
  
```

```

2B.  1 REM   T$,T1$ = TITLE (30 CHARACTERS)
      2 REM   T = BOOK NUMBER
      3 REM
      4 REM
      10 CLEAR 75
      20 OPEN "R",1,"BOOKS/TXT:1",30
      30 FIELD 1, 30 AS T$
      40 OPEN "O",2,"NZTITLES/TXT:1"
      50 FOR T = 1 TO 10
      60     GET 1
      70     IF T$ >= "N" THEN PRINT #2, T$
      80 NEXT T
      90 CLOSE 1,2
  
```

```

4A.  1 REM   N$,N1$ = NAME (20 CHARACTERS)
      2 REM   B,B$ = BALANCE (4 CHARACTERS)
      3 REM   A = ACCOUNT NUMBER
      10 CLEAR 100
      20 OPEN "R",1,"ACCOUNTS/TXT:1",24
      30 FIELD 1, 20 AS N$, 4 AS B$
      40 FOR A = 1 TO 25
      50     PRINT "THIS WILL BE ACCOUNT #";A
      60     INPUT "NAME";N1$ : LSET N$ = N1$
      70     INPUT "STARTING BALANCE";B : LSET B$ = MKS$(B)
      80     PUT 1
      90     PRINT
     100 NEXT A
     110 CLOSE 1
  
```

```

4B.  1 REM    N$ = NAME (20 CHARACTERS)
      2 REM    B$ = BALANCE (4 CHARACTERS)
      3 REM    A1 = ACCOUNT NUMBER
      4 REM    T$ = TRANSACTION FLAG
      5 REM    A2 = AMOUNT
      6 REM
      7 REM
      10 OPEN "R",1,"ACCOUNTS/TXT:1",24
      20 FIELD 1, 20 AS N$, 4 AS B$
      30 PRINT "ENTER A NEGATIVE NUMBER TO STOP"
      40 INPUT "ACCOUNT #";A1
      50     IF A1 < 0 THEN 140
      60     IF A > LOF(1) THEN PRINT "INVALID ACCOUNT #" : GOTO 40
      70     INPUT "DEPOSIT (D) OR WITHDRAWAL (W)";T$
      80     INPUT "AMOUNT";A2
      90     GET 1, A1
     100     IF T$ = "W" THEN LSET B$ = MKS$(CVS(B$) - A2)
                          ELSE LSET B$ = MKS$(CVS(B$) + A2)
     110     PUT 1, A1
     120     PRINT "TRANSACTION RECORDED"
     130 GOTO 40
     140 CLOSE 1

```

```

4C.  1 REM    N$ = NAME (20 CHARACTERS)
      2 REM    B$ = BALANCE (4 CHARACTERS)
      3 REM    A = ACCOUNT NUMBER
      4 REM
      5 REM
      10 OPEN "R",1,"ACCOUNTS/TXT:1",24
      20 FIELD 1, 20 AS N$, 4 AS B$
      30 FOR A = 1 TO 25
      40     GET 1
      50     IF CVS(B$) > 0 THEN 140
      60     PRINT
      70     PRINT "DEAR ";N$
      80     PRINT "      IT HAS COME TO OUR ATTENTION THAT YOUR"
      90     PRINT "ACCOUNT IS OVERDRAWN BY $";-CVS(B$);". PLEASE"
     100     PRINT "MAKE A DEPOSIT FOR THAT AMOUNT IMMEDIATELY."
     110     PRINT : PRINT TAB(40);"THANK YOU"
     120     PRINT TAB(40);"THE BANK"
     130     GOTO 210
     140     IF CVS(B$) < 500 THEN 210
     150     PRINT
     160     PRINT "DEAR ";N$
     170     PRINT "      YOU HAVE AN EXCEPTIONALLY LARGE ACCOUNT."
     180     PRINT "YOU WILL BE RECEIVING A TOASTER IN THE MAIL AS"
     190     PRINT "AN AWARD."
     200     PRINT : PRINT TAB(40);"THE BANK"
     210 NEXT A

```

```

6A.  1 REM    N$,N1$ = NAME (20 CHARACTERS)
      2 REM    F$,F1$ = FRATERNITY (20 CHARACTERS)
      3 REM    A,A$ = AGE (2 CHARACTERS)
      4 REM    S = STUDENT NUMBER
      5 REM
      6 REM
      10 CLEAR 100
      20 OPEN "R",1,"FRAT/TXT:1",42
      30 FIELD 1, 20 AS N$, 20 AS F$, 2 AS A$
      40 FOR S = 1 TO 30
      50     INPUT "NAME";N1$ : LSET N$ = N1$
      60     INPUT "FRATERNITY";F1$ : LSET F$ = F1$
      70     INPUT "AGE";A : LSET A$ = MKI$(A)
      80     PUT 1
      90     PRINT
      100 NEXT S
      110 CLOSE 1

```

```

6B.  1 REM    N$ = NAME (20 CHARACTERS)
      2 REM    F$ = FRATERNITY (20 CHARACTERS)
      3 REM    A$ = AGE (2 CHARACTERS)
      4 REM    S = STUDENT NUMBER
      5 REM
      6 REM
      10 CLEAR 100
      20 OPEN "R",1,"FRAT/TXT:1",42
      30 FIELD 1, 20 AS N$, 20 AS F$, 2 AS A$
      40 OPEN "D",2,"SIGMA/TXT:1"
      50 FOR S = 1 TO 30
      60     GET 1
      70     IF LEFT$(F$,9) = "SIGMA CHI" THEN PRINT #2, N$, CVI(A$)
      80 NEXT S
      90 CLOSE 1,2

```

```

6C.  1 REM    N$ = NAME (20 CHARACTERS)
      2 REM    F$ = FRATERNITY (20 CHARACTERS)
      3 REM    A$ = AGE (2 CHARACTERS)
      4 REM    S1 = STUDENT NUMBER
      5 REM    R,S = ROW, SEAT
      6 REM    R(14) = PREVIOUSLY PICKED STUDENT NUMBERS
      7 REM    C$(3,5) = CLASSROOM SEATING CHART
      8 REM
      9 REM
     10 CLEAR 750
     20 OPEN "R",1,"FRAT/TXT:1",42
     30 FIELD 1, 20 AS N$, 20 AS F$, 2 AS A$
     40 DIM R(14), C$(3,5)
     50 FOR R = 1 TO 3
     60     FOR S = 1 TO 5
     70         S1 = RND(30)
     80         FOR P = 1 TO (R-1)*5 + S-1
     90             IF S1 = R(P) THEN 70
    100             NEXT P
    110             R((R-1)*3+S) = S1
    120             GET 1, S1
    130             C$(R,S) = N$
    140         NEXT S
    150     NEXT R
    160 CLOSE 1
    170 FOR R = 1 TO 3
    180     PRINT "ROW";R;" : ",C$(R,1),C$(R,2)
    190     PRINT C$(R,3),C$(R,4),C$(R,5)
    200     PRINT
    210 NEXT R

```

```

8A.  1 REM    N$,N1$ = NAME (20 CHARACTERS)
      2 REM    SS$,S2$ = SOCIAL SECURITY NUMBER (11 CHARACTERS)
      3 REM    A$,A1$ = STREET ADDRESS (25 CHARACTERS)
      4 REM    C$,C1$ = CITY (15 CHARACTERS)
      5 REM    S$,S1$ = STATE (2 CHARACTERS)
      6 REM    Z$,Z1$ = ZIP CODE (5 CHARACTERS)
      7 REM    B,B$ = BALANCE (4 CHARACTERS)
      8 REM    A = ACCOUNT NUMBER
      9 REM
     10 REM
     20 CLEAR 500
     30 OPEN "R",1,"SAVINGS/TXT:1",82
     40 FIELD 1, 20 AS N$, 11 AS SS$, 25 AS A$, 15 AS C$, 2 AS S$,
        5 AS Z$, 4 AS B$
     50 FOR A = 1 TO 10
     60     PRINT "THIS WILL BE ACCOUNT #";A
     70     INPUT "NAME";N1$ : LSET N$ = N1$
     80     INPUT "SOCIAL SECURITY NUMBER";S2$ : LSET SS$ = S2$
     90     INPUT "STREET ADDRESS";A1$ : LSET A$ = A1$
    100     INPUT "CITY";C1$ : LSET C$ = C1$
    110     INPUT "STATE";S1$ : LSET S$ = S1$
    120     INPUT "ZIP CODE";Z1$ : LSET Z$ = Z1$
    130     INPUT "BALANCE";B : LSET B$ = MKS$(B)
    140     PUT 1
    150     PRINT
    160 NEXT A
    170 CLOSE 1

```

```

8B.  1 REM    N$,N1$ = NAME (20 CHARACTERS)
      2 REM    SS$,S2$ = SOCIAL SECURITY NUMBER (11 CHARACTERS)
      3 REM    A$,A1$ = STREET ADDRESS (25 CHARACTERS)
      4 REM    C$,C1$ = CITY (15 CHARACTERS)
      5 REM    S$,S1$ = STATE (2 CHARACTERS)
      6 REM    Z$,Z1$ = ZIP CODE (5 CHARACTERS)
      7 REM    B,B$ = BALANCE (4 CHARACTERS)
      8 REM    A = ACCOUNT NUMBER
      9 REM    O = OPTION
     10 REM
     11 REM
     20 CLEAR 500
     30 OPEN "R",1,"SAVINGS/TXT:1",82
     40 FIELD 1, 20 AS N$, 11 AS SS$, 25 AS A$, 15 AS C$, 2 AS S$,
        5 AS Z$, 4 AS B$
     50 INPUT "DEPOSITOR'S ACCOUNT NUMBER";A
     60 GET 1, A
     70 PRINT "NAME: ";N$
     80 PRINT , "(1) CHANGE NAME"
     90 PRINT , "(2) CHANGE SOCIAL SECURITY NUMBER"
    100 PRINT , "(3) CHANGE ADDRESS"
    110 PRINT , "(4) CHANGE BALANCE"
    120 PRINT
    130 INPUT "OPTION";O
    140 ON O GOTO 200,300,400,500
    150 REM
    160 REM
    200 REM    CHANGE NAME
    210 INPUT "NEW NAME";N1$ : LSET N$ = N1$
    220 GOTO 600
    230 REM
    240 REM
    300 REM    CHANGE SOCIAL SECURITY NUMBER
    310 PRINT "OLD SOCIAL SECURITY NUMBER: ";SS$
    320 INPUT "NEW SOCIAL SECURITY NUMBER";S2$ : LSET SS$ = S2$
    330 GOTO 600
    340 REM
    350 REM
    400 REM    CHANGE ADDRESS
    410 PRINT "OLD ADDRESS:"
    420 PRINT A$ : PRINT C$ : PRINT S1$ : PRINT Z$
    430 INPUT "NEW STREET ADDRESS";A1$ : LSET A$ = A1$
    440 INPUT "NEW CITY";C1$ : LSET C$ = C1$
    450 INPUT "NEW STATE";S1$ : LSET S$ = S1$
    460 INPUT "NEW ZIP CODE";Z1$ : LSET Z$ = Z1$
    470 GOTO 600
    480 REM
    490 REM
    500 REM    CHANGE BALANCE
    510 PRINT "OLD BALANCE: ";CVS(B$)
    520 INPUT "NEW BALANCE";B : LSET B$ = MKS$(B)
    530 REM
    540 REM
    600 REM    PLACE NEW INFORMATION IN FILE
    610 PUT 1, A
    620 CLOSE 1

```



```

8C.  1 REM   N$ = NAME (20 CHARACTERS)
      2 REM   SS$ = SOCIAL SECURITY NUMBER (11 CHARACTERS)
      3 REM   A$ = STREET ADDRESS (25 CHARACTERS)
      4 REM   C$ = CITY (15 CHARACTERS)
      5 REM   S$ = STATE (2 CHARACTERS)
      6 REM   Z$ = ZIP CODE (5 CHARACTERS)
      7 REM   B$ = BALANCE (4 CHARACTERS)
      8 REM   A = ACCOUNT NUMBER
      9 REM
     10 REM
     20 CLEAR 500
     30 OPEN "R",1,"SAVINGS/TXT:1",82
     40 FIELD 1, 20 AS N$, 11 AS SS$, 25 AS A$, 15 AS C$, 2 AS S$,
          5 AS Z$, 4 AS B$
     50 FOR A = 1 TO 10
     60     GET 1, A
     70     LSET B$ = MKS$(INT(100*(1.005*CVS(B$))+.5)/100)
     80     PUT 1, A
     90 NEXT A
    100 CLOSE 1

```

